

Lab05 : Les mesures et le Time Intelligence avec DAX

Dans cet atelier, vous allez **créer des mesures avec des expressions DAX** impliquant une manipulation de contexte de filtre.


Dans cet atelier, vous apprendrez à :

- Utilisez la fonction **CALCULATE()** pour manipuler le contexte du filtre
- Utiliser les fonctions Time Intelligence


Travailler avec le contexte de filtre

Dans cet exercice, vous allez créer des mesures avec des expressions DAX impliquant la manipulation du contexte de filtre.


1. Dans Power BI Desktop, dans la vue **Rapport**, créez une nouvelle page de rapport.



2. Sur la **page 3**, ajoutez un visuel de matrice.



3. Redimensionnez le visuel de la matrice pour remplir toute la page.
4. Pour configurer les champs visuels de la matrice, depuis le volet **Champs**, faites glisser la hiérarchie des régions **Region | Regions** et déposez-la dans le visuel.
5. Ajoutez également la colonne **Sales | Sales** de vente.
6. Pour développer toute la hiérarchie, en haut à droite du visuel de la matrice, cliquez deux fois sur l'icône en forme de double flèche fourchue.




On rappelle que la hiérarchie **Regions** comporte les niveaux **Group**, **Country**, and **Region**


7. Pour formater le visual, dans le volet **Visualisations**, sélectionnez le volet **Format**.



8. Dans la zone **Search**, saisissez **Stepped** ou bien **disposition échelonnée** (en français).



9. Définissez la propriété sur **Off**.



10. Vérifiez que le visuel de la matrice comporte désormais quatre en-têtes de colonne.

[Retour au rapport](#)

Group	Country	Region	Somme de Sales
Europe	France	France	\$4 527 839,69
		Total	\$4 527 839,69
	Germany	Germany	\$1 877 743,39
		Total	\$1 877 743,39
	United Kingdom	United Kingdom	\$3 883 042,96
North America	Canada	Canada	\$13 875 632,96
		Total	\$13 875 632,96
	United States	Central	\$7 633 386,86
		Northeast	\$6 715 354,39
		Northwest	\$12 004 822,45
		Southeast	\$7 638 606,55
Pacific	Australia	Southwest	\$18 001 116,10
		Total	\$51 993 286,35
	Australia	Australia	\$1 391 024,85
		Total	\$1 391 024,85
	Total	Total	\$77 548 570,20

Chez Adventure Works, les régions de vente sont organisées en groupes, pays et régions. Tous les pays, à l'exception des États-Unis, n'ont qu'une seule région, qui porte le nom du pays. Les États-Unis étant un territoire de vente si vaste, il est divisé en cinq régions de vente.

Manipuler le contexte du filtre

Dans cette tâche, vous allez créer plusieurs mesures avec des expressions DAX qui utilisent la fonction **CALCULATE()** pour manipuler le contexte du filtre.

11. Ajoutez une mesure à la table **Sales**, basée sur l'expression suivante :

```
Sales All Region =
CALCULATE(SUM(Sales[Sales]), REMOVEFILTERS(Region))
```

La fonction **CALCULATE()** est une fonction puissante utilisée pour manipuler le contexte du filtre. Le premier argument prend une expression ou une mesure (une mesure est juste une expression nommée). Les arguments suivants permettent de modifier le contexte du filtre.

La fonction **REMOVEFILTERS()** supprime les filtres actifs. Elle peut prendre soit aucun argument, soit une table, une colonne ou plusieurs colonnes comme argument.

Dans cette formule, la mesure évalue la somme de la colonne **Sales** dans un contexte de filtre modifié, qui supprime tous les filtres appliqués aux colonnes de la table **Region**.

12. Ajoutez la mesure **Sales All Region** au visuel de matrice.

[Retour au rapport](#)

Group	Country	Region	Somme de Sales	Sales All Region
Europe	France	France	\$4 527 839,69	77 548 570,20
		Total	\$4 527 839,69	77 548 570,20
	Germany	Germany	\$1 877 743,39	77 548 570,20
		Total	\$1 877 743,39	77 548 570,20
	United Kingdom	United Kingdom	\$3 883 042,96	77 548 570,20
North America	Canada	Canada	\$13 875 632,96	77 548 570,20
		Total	\$13 875 632,96	77 548 570,20
	United States	Central	\$7 633 386,86	77 548 570,20
		Northeast	\$6 715 354,39	77 548 570,20
		Northwest	\$12 004 822,45	77 548 570,20
		Southeast	\$7 638 606,55	77 548 570,20
		Southwest	\$18 001 116,10	77 548 570,20
		Total	\$51 993 286,35	77 548 570,20
	Australia	Australia	\$1 391 024,85	77 548 570,20
		Total	\$1 391 024,85	77 548 570,20
	Total		\$1 391 024,85	77 548 570,20
	Total		\$77 548 570,20	77 548 570,20

13. Notez que la mesure **Sales All Region** calcule le total de toutes les ventes régionales pour chaque région, pays (sous-total) et groupe (sous-total).

La nouvelle mesure n'a pas encore produit de résultat utile. Lorsque les ventes d'un groupe, d'un pays ou d'une région sont divisées par cette valeur, cela produit un ratio utile appelé "**pourcentage du total général**".

14. Dans le volet **Champs**, assurez-vous que la mesure **Sales All Region** est sélectionnée, puis dans la barre de formule, remplacez le nom de la mesure et la formule par la formule suivante :

Conseil : Pour remplacer la formule existante, copiez d'abord l'extrait de code. Ensuite, cliquez dans la barre de formule et appuyez sur **Ctrl+A** pour sélectionner tout le texte. Ensuite, appuyez sur **Ctrl + V** pour coller l'extrait de code afin d'écraser le texte sélectionné. Appuyez ensuite sur **Entrée**.

```
Sales % All Region =
DIVIDE(
    SUM(Sales[Sales]),
    CALCULATE(
        SUM(Sales[Sales]),
        REMOVEFILTERS(Region)
    )
)
```

La mesure a été renommée pour refléter avec précision la formule mise à jour. La fonction **DIVIDE()** divise la mesure **Sales** (non modifiée par le contexte de filtre) par la mesure **Sales** dans un contexte modifié, ce qui supprime tous les filtres appliqués à la table **Region**.

15. Dans le visuel de la matrice, notez que la mesure a été renommée et qu'une valeur différente apparaît désormais pour chaque groupe, pays et région.
16. Formatez la mesure **Sales % All Region** sous la forme d'un pourcentage avec deux décimales.
17. Dans le visuel de la matrice, passez en revue les valeurs de mesure **Sales % All Region**.

Group	Country	Region	Somme de Sales	Sales % All Region
Europe	France	France	\$4 527 839,69	5,84
		Total	\$4 527 839,69	5,84
	Germany	Germany	\$1 877 743,39	2,42
		Total	\$1 877 743,39	2,42
	United Kingdom	United Kingdom	\$3 883 042,96	5,01
		Total	\$3 883 042,96	5,01
	Total		\$10 288 626,04	13,27
North America	Canada	Canada	\$13 875 632,96	17,89
		Total	\$13 875 632,96	17,89
	United States	Central	\$7 633 386,86	9,84
		Northeast	\$6 715 354,39	8,66
		Northwest	\$12 004 822,45	15,48
		Southeast	\$7 638 606,55	9,85
		Southwest	\$18 001 116,10	23,21
		Total	\$51 993 286,35	67,05
Total		\$65 868 919,31	84,94	
Pacific	Australia	Australia	\$1 391 024,85	1,79
		Total	\$1 391 024,85	1,79
	Total		\$1 391 024,85	1,79
Total			\$77 548 570,20	100,00

18. Ajoutez une autre mesure à la table **Sales**, basée sur l'expression suivante, et formatez-la en pourcentage :

```
Sales % Country =
DIVIDE (
    SUM(Sales[Sales]),
    CALCULATE (
        SUM(Sales[Sales]),
        REMOVEFILTERS (Region[Region])
    )
)
```

19. Notez que la formule de mesure **Sales % Country** diffère légèrement de la formule de mesure **Sales % All Region**.

La différence est que le dénominateur modifie le contexte du filtre en supprimant les filtres de la colonne **Region** de la table **Region**, et non de toutes les colonnes de la table **Region**. Cela signifie que tous les filtres appliqués aux colonnes de groupe ou de pays sont conservés. Il obtiendra un résultat qui représente les ventes en pourcentage du pays.

20. Ajoutez la mesure **Sales % Country** au visuel de la matrice.
21. Notez que seules les régions des États-Unis produisent une valeur qui n'est pas de 100 %.

Group	Country	Region	Somme de Sales	Sales % All Region	Sales % Country
Europe	France	France	\$4 527 839,69	5,84 %	100,00 %
		Total	\$4 527 839,69	5,84 %	100,00 %
	Germany	Germany	\$1 877 743,39	2,42 %	100,00 %
		Total	\$1 877 743,39	2,42 %	100,00 %
	United Kingdom	United Kingdom	\$3 883 042,96	5,01 %	100,00 %
		Total	\$3 883 042,96	5,01 %	100,00 %
	Total		\$10 288 626,04	13,27 %	100,00 %
North America	Canada	Canada	\$13 875 632,96	17,89 %	100,00 %
		Total	\$13 875 632,96	17,89 %	100,00 %
	United States	Central	\$7 633 386,86	9,84 %	14,68 %
		Northeast	\$6 715 354,39	8,66 %	12,92 %
		Northwest	\$12 004 822,45	15,48 %	23,09 %
		Southeast	\$7 638 606,55	9,85 %	14,69 %
	Southwest	\$18 001 116,10	23,21 %	34,62 %	
Total		\$51 993 286,35	67,05 %	100,00 %	
Total			\$65 868 919,31	84,94 %	100,00 %
Pacific	Australia	Australia	\$1 391 024,85	1,79 %	100,00 %
		Total	\$1 391 024,85	1,79 %	100,00 %
	Total		\$1 391 024,85	1,79 %	100,00 %
Total			\$77 548 570,20	100,00 %	100,00 %

Vous vous souviendrez peut-être que seuls les États-Unis ont plusieurs régions. Tous les autres pays forment une seule région, ce qui explique qu'ils soient tous à 100 %.

22. Pour améliorer la lisibilité de cette mesure dans le visuel, remplacez la mesure **Sales % Country** par cette formule améliorée.

```
Sales % Country =
IF (
    ISINSCOPE (Region[Region]),
    DIVIDE (
        SUM (Sales[Sales]),
        CALCULATE (
            SUM (Sales[Sales]),
            REMOVEFILTERS (Region[Region])
        )
    )
)
```

Intégrée à la fonction **IF()**, la fonction **ISINSCOPE()** est utilisée pour tester si la colonne de région est le niveau dans une hiérarchie de niveaux. Lorsque c'est **True**, la fonction **DIVIDE()** est évaluée. L'absence d'une fausse partie signifie qu'un blanc est renvoyé lorsque la colonne de région n'est pas dans la portée.

23. Notez que la mesure **Sales % Country** ne renvoie désormais une valeur que lorsqu'une région est dans la portée.

Group	Country	Region	Sales	Sales % All Region	Sales % Country
Europe	France	France	\$4,527,840	5.84 %	100.00 %
		Total	\$4,527,840	5.84 %	
	Germany	Germany	\$1,877,743	2.42 %	100.00 %
		Total	\$1,877,743	2.42 %	
	United Kingdom	United Kingdom	\$3,883,043	5.01 %	100.00 %
		Total	\$3,883,043	5.01 %	

24. Ajoutez une autre mesure à la table **Sales**, basée sur l'expression suivante, et formatez-la en pourcentage :

```
Sales % Group =
DIVIDE (
    SUM (Sales[Sales]),
    CALCULATE (
        SUM (Sales[Sales]),
        REMOVEFILTERS (
```

```

Region[Region],
Region[Country]
)
)
)

```

Pour obtenir des ventes en pourcentage du groupe, deux filtres peuvent être appliqués pour supprimer efficacement les filtres sur deux colonnes.

25. Ajoutez la mesure **Sales % Group** au visuel de la matrice.

Group	Country	Region	Somme de Sales	Sales % All Region	Sales % Country	Sales % Group
Europe	France	France	\$4 527 839.69	5,84 %	100,00 %	44,01 %
		Total	\$4 527 839.69	5,84 %		44,01 %
	Germany	Germany	\$1 877 743.39	2,42 %	100,00 %	18,25 %
		Total	\$1 877 743.39	2,42 %		18,25 %
	United Kingdom	United Kingdom	\$3 883 042.96	5,01 %	100,00 %	37,74 %
		Total	\$3 883 042.96	5,01 %		37,74 %
	Total		\$10 288 626.04	13,27 %		100,00 %
North America	Canada	Canada	\$13 875 632.96	17,89 %	100,00 %	21,07 %
		Total	\$13 875 632.96	17,89 %		21,07 %
	United States	Central	\$7 633 386.86	9,84 %	14,68 %	11,59 %
		Northeast	\$6 715 354.39	8,66 %	12,92 %	10,20 %
		Northwest	\$12 004 822.45	15,48 %	23,09 %	18,23 %
		Southeast	\$7 638 606.55	9,85 %	14,69 %	11,60 %
		Southwest	\$18 001 116.10	23,21 %	34,62 %	27,33 %
		Total	\$51 993 286.35	67,05 %		78,93 %
	Total		\$65 868 919.31	84,94 %		100,00 %
	Pacific	Australia	Australia	\$1 391 024.85	1,79 %	100,00 %
Total			\$1 391 024.85	1,79 %		100,00 %
Total			\$1 391 024.85	1,79 %		100,00 %
Total			\$77 548 570.20	100,00 %		100,00 %

26. Pour améliorer la lisibilité de cette mesure dans le visuel, remplacez la mesure **Sales % Group** par cette formule améliorée.

```


Sales % Group =
IF (
    ISINSCOPE (Region[Region])
    || ISINSCOPE (Region[Country]),
    DIVIDE (
        SUM (Sales[Sales]),
        CALCULATE (
            SUM (Sales[Sales]),
            REMOVEFILTERS (
                Region[Region],
                Region[Country]
            )
        )
    )
)

```

27. Notez que la mesure **Sales % Group** ne renvoie désormais une valeur que lorsqu'une région ou un pays est dans la portée.

Group	Country	Region	Somme de Sales	Sales % All Region	Sales % Country	Sales % Group
Europe	France	France	\$4 527 839,69	5,84	100,00	44,01
		Total	\$4 527 839,69	5,84		44,01
	Germany	Germany	\$1 877 743,39	2,42	100,00	18,25
		Total	\$1 877 743,39	2,42		18,25
	United Kingdom	United Kingdom	\$3 883 042,96	5,01	100,00	37,74
		Total	\$3 883 042,96	5,01		37,74
North America	Canada	Canada	\$13 875 632,96	17,89	100,00	21,07
		Total	\$13 875 632,96	17,89		21,07
	United States	Central	\$7 633 386,86	9,84	14,68	11,59
		Northeast	\$6 715 354,39	8,66	12,92	10,20
		Northwest	\$12 004 822,45	15,48	23,09	18,23
		Southeast	\$7 638 606,55	9,85	14,69	11,60
		Southwest	\$18 001 116,10	23,21	34,62	27,33
	Total	\$51 993 286,35	67,05		78,93	
	Total		\$65 868 919,31	84,94		
	Pacific	Australia	Australia	\$1 391 024,85	1,79	100,00
Total			\$1 391 024,85	1,79		100,00
Total		\$1 391 024,85	1,79			
Total			\$77 548 570,20	100,00		

28. Dans la vue **Modèle**, placez les trois nouvelles mesures dans un dossier d'affichage nommé **Ratios**.



29. Enregistrez le fichier Power BI Desktop.

Les mesures ajoutées à la table **Sales** ont modifié le contexte du filtre pour obtenir une navigation hiérarchique.


Travailler avec Time Intelligence

Dans cet exercice, vous allez créer une mesure des ventes depuis le début de l'année (**YTD**) et une mesure de la croissance des ventes d'une année sur l'autre (**YoY**).

30. Dans la vue **Rapport**, sur la **page 2**, notez le visuel de la matrice qui affiche diverses mesures avec les années et les mois regroupés sur les lignes.

31. Ajoutez une mesure à la table **Sales**, basée sur l'expression suivante et mise en forme avec zéro décimale :

```
Sales YTD =  
TOTALYTD(SUM(Sales[Sales]), 'Date'[Date], "6-30")
```



La fonction **TOTALYTD()** évalue une expression, dans ce cas la somme de la colonne **Sales**, sur une colonne de date donnée **Date**. La colonne de date doit appartenir à une table de dates marquée comme table de dates. La fonction peut également prendre un troisième argument facultatif représentant la dernière date d'une année. L'absence de cette date signifie que le 31 décembre est le dernier jour de l'année. Pour Adventure Works, juin du dernier mois de leur année, et donc "6-30" est utilisé.

32. Ajoutez le champ **Sales** et la mesure **Sales YTD** au visuel de la matrice.

33. Remarquez l'accumulation des valeurs des ventes au cours de l'année.

Year	Avg Price	Median Price	Min Price	Max Price	Order Lines	Orders	Somme de Sales	Sales YTD
FY2018	748,68 €	419,46 €	4,75 €	2 146,96 €	8 459	739	\$16 429 042,60	\$16 429 042,60
2017 juil	655,59 €	419,46 €	5,19 €	2 146,96 €	352	38	\$489 328,40	\$489 328,40
2017 août	758,93 €	419,46 €	4,75 €	2 146,96 €	785	75	\$1 540 072,02	\$2 029 400,42
2017 sept	741,85 €	419,46 €	5,19 €	2 146,96 €	593	60	\$1 166 332,45	\$3 195 732,87
2017 oct	677,45 €	419,46 €	5,19 €	2 146,96 €	499	40	\$844 832,71	\$4 040 565,58
2017 nov	752,31 €	419,46 €	5,01 €	2 146,96 €	1 106	90	\$2 325 754,91	\$6 366 320,49
2017 déc	734,58 €	419,46 €	5,01 €	2 146,96 €	803	63	\$1 703 435,09	\$8 069 755,58
2018 janv	808,94 €	419,46 €	5,19 €	2 146,96 €	377	40	\$713 229,54	\$8 782 985,12
2018 févr	896,80 €	419,46 €	5,01 €	2 146,96 €	866	79	\$1 900 794,46	\$10 683 779,58
2018 mars	863,54 €	419,46 €	5,19 €	2 146,96 €	653	64	\$1 455 280,27	\$12 139 059,85
2018 avr	732,25 €	419,46 €	5,19 €	2 146,96 €	494	37	\$883 010,98	\$13 022 070,83
2018 mai	761,30 €	419,46 €	4,75 €	2 146,96 €	1 112	85	\$2 269 719,87	\$15 291 790,70
2018 juin	552,95 €	419,46 €	4,75 €	2 146,96 €	819	68	\$1 137 251,90	\$16 429 042,60
FY2019	397,81 €	202,33 €	4,32 €	1 466,01 €	21 670	1 255	\$27 979 779,53	\$27 979 779,53
FY2020	392,12 €	200,05 €	1,33 €	1 466,01 €	27 722	1 622	\$33 139 748,07	\$33 139 748,07
Total	446,39 €	214,24 €	1,33 €	2 146,96 €	57 851	3 616	\$77 548 570,20	

La fonction **TOTALYTD()** effectue la manipulation du filtre, en particulier la manipulation du filtre temporel. Par exemple, pour calculer les ventes YTD pour septembre 2017 (le troisième mois de l'exercice), tous les filtres de la table **Date** sont supprimés et remplacés par un nouveau filtre de dates commençant au début de l'année (1er juillet 2017) et s'étendant jusqu'à la dernière date de la période de dates en contexte (30 septembre 2017).

34. Ajoutez une mesure supplémentaire à la table **Sales**, basée sur l'expression suivante :

```
Sales YoY Growth =
VAR SalesPriorYear =
    CALCULATE (
        SUM(Sales[Sales]),
        PARALLELPERIOD (
            'Date'[Date],
            -12,
            MONTH
        )
    )
RETURN
    SalesPriorYear
```

La formule de mesure **Sales YoY Growth** déclare une variable. Les variables peuvent être utiles pour simplifier la logique de la formule, et plus efficaces lorsqu'une expression doit être évaluée plusieurs fois dans la formule (ce qui sera le cas pour la logique de croissance YoY). Les variables sont déclarées par un nom unique et l'expression de la mesure doit ensuite être sortie après le mot-clé **RETURN**.

La variable **SalesPriorYear** se voit attribuer une expression qui calcule la somme de la colonne **Sales** dans un contexte modifié qui utilise la fonction **PARALLELPERIOD()** pour reculer de 12 mois à partir de chaque date dans le contexte du filtre.

35. Ajoutez la mesure **Sales YoY Growth** au visuel de la matrice.

36. Notez que la nouvelle mesure renvoie BLANC pour les 12 premiers mois (car aucune vente n'a été enregistrée avant l'exercice 2017).

37. Notez que la valeur de la mesure **Sales YoY Growth** pour **2018 Jul** est la valeur **Sales** pour **2017 Jul**.

Sales	Sales YTD	Sales YoY Growth
\$16,429,043	\$16,429,043	
\$489,328	\$489,328	
\$1,540,072	\$2,029,400	
\$1,166,333	\$3,195,733	
\$844,833	\$4,040,566	
\$2,325,755	\$6,366,320	
\$1,703,435	\$8,069,756	
\$713,230	\$8,782,985	
\$1,900,794	\$10,683,780	
\$1,455,280	\$12,139,060	
\$883,011	\$13,022,071	
\$2,269,720	\$15,291,791	
\$1,137,252	\$16,429,043	
\$27,979,780	\$27,979,780	\$16,429,042.6
\$2,411,559	\$2,411,559	\$489,328.4
\$3,615,914	\$6,027,473	\$1,540,072.02

Maintenant que la « partie difficile » de la formule a été testée, vous pouvez remplacer la mesure par la formule finale qui calcule le résultat de la croissance.

38. Pour compléter la mesure, remplacez la mesure **Sales YoY Growth** par cette formule, en la formatant en pourcentage avec deux décimales :

```
Sales YoY Growth =
VAR SalesPriorYear =
    CALCULATE(
        SUM(Sales[Sales]),
        PARALLELPERIOD(
            'Date'[Date],
            -12,
            MONTH
        )
    )
RETURN
    DIVIDE(
        (SUM(Sales[Sales]) - SalesPriorYear),
        SalesPriorYear
    )
```


39. Dans la formule, dans la clause **RETURN**, notez que la variable est référencée deux fois.

40. Vérifiez que la croissance annuelle pour **juillet 2018** est de **392,83 %**.

Sales	Sales YTD	Sales YoY Growth
\$16,429,043	\$16,429,043	
\$489,328	\$489,328	
\$1,540,072	\$2,029,400	
\$1,166,332	\$3,195,733	
\$844,833	\$4,040,566	
\$2,325,755	\$6,366,320	
\$1,703,435	\$8,069,756	
\$713,230	\$8,782,985	
\$1,900,794	\$10,683,780	
\$1,455,280	\$12,139,060	
\$883,011	\$13,022,071	
\$2,269,720	\$15,291,791	
\$1,137,252	\$16,429,043	
\$27,979,780	\$27,979,780	70.31 %
\$2,411,559	\$2,411,559	392.83 %
\$3,615,914	\$6,027,473	134.79 %


Cela signifie que les ventes de **juillet 2018 (2 411 559 \$)** représentent une amélioration de près de 400 % (presque 4x) par rapport aux ventes réalisées à la même période de l'année précédente (489 328 \$).

41. Dans la vue **Modèle**, placez les deux nouvelles mesures dans un dossier d'affichage nommé **Time Intelligence**.



Dans cette tâche, vous terminerez le laboratoire.

42. Pour nettoyer la solution prête pour le développement du rapport, en bas à gauche, cliquez avec le bouton droit sur l'onglet **Page 2**, puis sélectionnez **Supprimer** la page.



43. Lorsque vous êtes invité à supprimer la page, cliquez sur **Supprimer**.
 44. Supprimez également **la page 3**.
 45. Sur la page restante, pour effacer la page, sélectionnez le visuel du tableau, puis appuyez sur la touche **Suppr**.
 46. Enregistrez le fichier Power BI Desktop.